VLB ARRAY MEMO No. 321

SINGLE SIDEBAND FRINGE ROTATION

Larry R. D'Addario
4 February 1984

When fringe rotation is done at baseband, undesired products that do not contribute to the cross correlation, but that do contribute noise, are generated. This is because multiplication of the signal by a sinusoid at the fringe frequency shifts each component both up and down by the fringe frequency, and only one of these is correct. The undesired product can be suppressed if the signal is multiplied by two fringe-frequency sinusoids in quadrature; if one of these products is passed through a 90-degree phase shift filter and then added to (or subtracted from) the other product, the undesired part is cancelled. This phase shifting and recombining can be done before or after cross correlation with the signal from another antenna, but in the latter case twice as much cross correlation must be done. To do it before cross correlation, a 90-degree phase shift filter must be built. One way of approximating such a filter is discussed in this memo.

Let $s(t)$ be the signal to be fringe-rotated. Formally, we can write

$$s(t) = s_+(t) + s_-(t)$$

where $s_+(t)$ is the positive frequency part and $s_-(t)$ is the negative frequency part (they are complex conjugates, and s_+ is one half of the analytic signal of s). This is convenient because

$$2 \cos(wt) = \exp(jwt) + \exp(-jwt)$$

and

$$2 \sin(wt) = -j\, [\exp(jwt) - \exp(-jwt)].$$

Thus, the product of $s(t)$ with each of the latter functions can be written

$$C(t) = s_+\exp(jwt) + s_-\exp(-jwt) + s_+\exp(-jwt) + s_-\exp(jwt)$$

and

$$S(t) = -j[s_+\exp(jwt) - s_-\exp(-jwt) + s_+\exp(-jwt) - s_-\exp(jwt)].$$

If we assume that a positive frequency shift is desired, then the first two terms in each equation are the desired products and the others are the undesired ones. Also, the first and third terms contain only positive frequencies and the others contain only negative frequencies. Now suppose that $S(t)$ is passed through a filter whose impulse response is $H(t)$, and that the result is added to $C(t)$, giving:

$$\begin{aligned} s'(t) &= C(t) + S(t)*H(t). \\ &= [s_+\exp(jwt)]*[1 - jH(t)] + [s_-\exp(-jwt)]*[1 + jH(t)] \\ &\quad + [s_+\exp(-jwt)]*[1 + jH(t)] + [s_-\exp(jwt)]*[1 - jH(t)] \end{aligned}$$

where * represents convolution. It should be apparent from the above that the undesired terms cancel if H has a Fourier transform that is $+j$ for positive frequencies and $-j$ for negative frequencies. This makes it, as expected, a 90 degree phase shifter or Hilbert transform filter.

In the last equation, the first two terms are the desired ones and the others are undesired; the first and third terms contain only positive frequencies and the others only negative frequencies. Considering only positive frequencies, we find that the signal to noise ratio is reduced, relative to ideal filtering, by the factor

$$\mathrm{SNR} = \frac{|1 - jh(f)|}{\sqrt{[1 - jh(f)]^2 + [1 + jh(f)]^2}} \quad .$$

where $h(f)$ is the Fourier transform of $H(t)$. The denominator terms are the contributions from the desired and undesired products, which are independent and hence add in power.

Next, suppose that $s(t)$ and $s'(t)$ are discrete time (sampled) signals; i.e., $s(t)$ is only available at $t = i/f_s$, for integer i, and we are only interested in $s'(t)$ for the same values of t. This causes no loss of information if the maximum frequency in $s(t)$ is less than $f_s/2$. Then $H(t)$ can be band-limited to $f_s/2$ also, and only its samples are of interest. It turns out that the ideal filter has the following discrete impulse response:

... -1/7, 0, -1/5, 0, -1/3, 0, -1, 0, 1, 0, 1/3, 0, 1/5, 0, 1/7,...

By truncating this symmetrically about the center, and by inserting a delay of half the truncated length, we obtain a realizable filter with a finite impulse response (FIR). The use of such a filter in a fringe rotator is illustrated in Figure 1.

Figure 3 shows the calculated frequency responses of fringe rotators based on FIR filters with 4, 6, and 8 "taps"; for N taps one obtains a filter of length $2N+1$, because alternate samples of the impulse response are zero. The effect on the signal to noise ratio, relative to a perfect filter, is also plotted. At the band edges, no suppression of the undesired product is obtained, and so the SNR is reduced by $\sqrt{2}$; but for even the simplest filter, the loss is less than 10% over 90% of the band, and is less than 1% over 80% of the band. For continuum work, the reduction in effective bandwidth causes a loss which is also given in Figure 3; normally, the effective bandwidth will also be reduced by other filters in the system, in which case the percentage loss from this filter will be less.

The calculated SNR loss factors are ratios of signal amplitude to noise standard deviation for a single signal. If similar fringe rotators are installed on both signals of a baseline, the loss factor should be squared.

This calculation has not specified whether the input signals are continuous in magnitude or quantized to a finite number of levels; thus, the results apply equally well in either case. But we have not considered quantization of the fringe frequency sinusoids, nor re-quantization of the output signal, both of which would be required in a digital implementation. These will be studied in a later memo.



FIGURE 1: Fringe rotator based on an FIR filter with 2N taps. Boxes labeled kD are delays of k sample times.

]

]

FIGURE 2: Listing of BASIC program used to generate plots of Fig. 3.

```
START=16385  END=17280  LENGTH=895
1  REM ===FIR FILTER CALCULATIONS===
2  REM  FOR VLBA FRINGE ROTATORS
3  REM  840209 LRD.
100 DF = .002:DX = 279 / .5:DY = 191
105 P4 = 6.2831853: REM  2*PI
110  INPUT "HALF NUMBER OF TAPS: ";N
120  FOR I = 1 TO N: PRINT "COEFF "I"=";: INPUT "";A(I): NEXT
130  GOSUB 1000: REM  MAKE BORDER
200 SI = 0:NO = 0: FOR F = 0 TO .5 STEP DF
210 W = P4 * F:J = 0
220  FOR I = 1 TO N:J = J + A(I) *  SIN (W * (I + I - 1)): NEXT
230 X = (1 - J) / (1 + J):SNR = 1 /  SQR (1 + X * X)
235 X =  INT (F * DX + .5)
240  HPLOT X,191.5 - (J + 1) / 4 * DY
250  HPLOT X,191.5 - (SNR - .70) / .3 * DY
260 SI = SI + J:NO = NO + J * J: NEXT
300  PRINT  CHR$ (4)"PR#1": PRINT
310  PRINT "FOR "2 * N" TAPS:"
320  FOR I = 1 TO N: PRINT A(I),: NEXT
330  PRINT : PRINT "CONTINUUM SNR= "(.5 + DF * SI) /  SQR (.5 + DF * NO)
340  PRINT  CHR$ (4)"PR#0"
400  GET A$
410  IF A$ = "T" THEN  POKE  - 16303,0: GOTO 400
420  IF A$ = "G" THEN  POKE  - 16304,0: GOTO 400
430  IF A$ = "E" THEN  TEXT : END : GOTO 400
440  IF A$ = "C" THEN  PRINT : POKE 1145,49: CALL  - 16038: GOTO 400
450  GOTO 400
1000 :
1005  REM   *** SET UP SCREEN ***
1010 :
1020  HGR : HCOLOR= 3: POKE  - 16304,0: POKE  - 16302,0: POKE  - 16297,0
1030  HPLOT 0,0 TO 279,0 TO 279,191 TO 0,191 TO 0,0
1040  FOR J = 19.1 TO 172 STEP 19.1: HPLOT 0,J TO 3,J: HPLOT 276,J TO 279,J:
 NEXT
1050  FOR J = 27.9 TO 252 STEP 27.9: HPLOT J,0 TO J,3: HPLOT J,188 TO J,191:
 NEXT
1060  RETURN
```



FIGURE 3: Performance of fringe rotators using FIR filters with 4, 6, and 8 taps.